

ANDRE GARZIA

LIVECODE
ADVANCED
APPLICATION
ARCHITECTURE

From HOBBYIST to PRO

LiveCode Advanced Application Architecture

From HOBBYIST to PRO

Andre Garzia

This book is for sale at <http://leanpub.com/livecodeapparchitecture>

This version was published on 2018-11-19



This is a [Leanpub](#) book. Leanpub empowers authors and publishers with the Lean Publishing process. [Lean Publishing](#) is the act of publishing an in-progress ebook using lightweight tools and many iterations to get reader feedback, pivot until you have the right book and build traction once you do.

© 2013 - 2018 Andre Garzia

Tweet This Book!

Please help Andre Garzia by spreading the word about this book on [Twitter](#)!

The suggested tweet for this book is:

[I am learning how to organize my code and create easy to maintain apps with LiveCode](#)

The suggested hashtag for this book is [#livecode](#).

Find out what other people are saying about the book by clicking on this link to search for this hashtag on Twitter:

[#livecode](#)

Also By **Andre Garzia**

[Guia Rápido de Desenvolvimento para Firefox OS](#)

[Quick Guide For Firefox OS App Development](#)

[Building Games for Firefox OS](#)

[Quick Guide For Firefox OS App Development](#)

To Elisangela Mendonça de Andrade Garzia

Contents

Disclaimer	i
Intended audience	i
Thanks	i
on openBook	1
From hobbyist to pro	1
If I do my job right, I will	2
This book roadmap	2
Code Conventions	3
How to name things	3
How to document your code	6
An introduction to MVC	8
MVC in 20 seconds	9
Summary	10
Models	11
What goes into the model and what doesn't	11
Building a sane API	11
Practical: Building an address book model	12
Summary	18
Controllers	19
Controllers are like transit cops	19
What goes into the controller and what doesn't	19
What shouldn't be on a controller	20
Practical: Building an address book controller	20
The publish and subscribe pattern	21
Implementing our controller	22
The controller script	23
Summary	25
Views	26
Practical tips for interface design	26
What goes into the view and what doesn't	26

CONTENTS

Multiple views are the key to cross-platform applications	27
Practical: Building views for the address book application	27
How our view work?	28
The contacts list	29
The details view	32
on closeBook	36

Disclaimer

This is an early release. Send feedback to feedback@andregarzia.com¹.

The content of this book will be expanded in the next months, I see this book as a living thing, much like LiveCode IDE.

This is the “minimum amount of book” I consider useful enough to make a release. I hope you all enjoy and benefit from it.

Intended audience

[LiveCode](#)² developers who want to learn a more structured way of organizing and reusing their code. This book assumes beginner or intermediate knowledge of LiveCode. If you joined our beloved community recently there are other [resources that will help you get started](#)³.

Thanks

Thanks to every single member of the LiveCode community. I remember being a newbie in 2004 and joining this awesome crew full of questions and doubts. You all received me and helped shape my career and life. If I were to thank people personally I would need Heather to give me a dump of all use-list subscribers from 2004 to 2018 (oh my, it has been more than ten years...).


You people are just too awesome!

Since this is a book, I want to specify a very special friend here in this section: Dan “the man” Shafer. From my early days till today, Dan was always someone I could look for inspiration. I remember meeting him in 2004 and thinking that one day I would like to write books as well. This is my third programming book, I might be getting somewhere...

¹<mailto:feedback@andregarzia.com>

²<http://livecode.com>

³<https://livecode.com/resources/guides/>



Revolution at the speed of thought

I still consider myself an inventive user more than a developer anyway and the book above still on my shelf...

on openBook

I've been a speaker on LiveCode conferences multiple times speaking about many different topics such as web development, game design and application organization. The talk that received the best feedback was the one about application organization during RunRevLive 2013. I remember the smiles from the audience as I moved away from the speaker microphone. That mental image and the the kind words received as I left the stage kept rumbling on my mind for a while as I tried to understand why that session was so well received while other sessions were less impressive, then one day late at night, I grokked it!

Even though *Web development* and *Game design* are very interesting topics, they are not the most common activities in our community. When I gave sessions on web development, those doing that kind of work really enjoyed the session but those that don't do anything related to the Web were not all that impressed. The same thing happened with game design. Still, application organization is a topic that affects everyone. All LiveCoders, from desktop application developers, to WebMasters, to Game Designers, start with the same felling: "*LiveCode is so easy that I don't really need to organize my stuff!*" and then, like instantaneous karma, the regret from not having a defined methodology comes right back at us while we wonder: "*Where did I placed that handler?*".

We enjoy throwing stacks around while we're learning the language, I have stacks so badly designed that they try to hurt you when you look at the code. Those stacks from *the early spaghetti days* are *a huge contrast to what I try to build now*. My objective with this book is to codify what I've learned as I figured out that my code needed to be better organized and easier to maintain if I were to win in the freelance market. Let us label this improvement of skills **going from hobbyist to pro**.

From hobbyist to pro

If you pick the average sample code contribution by new developers that are learning LiveCode as their first computer programming language and compare it with what the seasoned developers who have business built around the language and have been shipping applications for a number of years, you will notice a huge difference in both code clarity and organization.

Going from hobbyist to pro is not the act of building a business or shipping commercial software but the act of working using best practices. This book is about exactly this, picking knowledge from seasoned developers and moving so that we stop shipping spaghetti code.

I don't claim that the techniques presented here were invented by me and lots of seasoned developers will recognize their style in one part or another. This book has many techniques in it and while I try to present them as a solid and cohesive architecture, remember that you can tweak things to suit your application. You should tailor the methodology shown here to your own needs and not the other way around. With that in mind, lets review the books objective.

If I do my job right, I will

Present a solid architecture for building application using LiveCode using an MVC-like approach. By the end of the book (if you read everything) you should be able to:

- Name your variable and handlers with meaningful names.
- Correctly place code in stacks.
- Divide your application into logical units such as models, views, controllers and libraries.
- Have a clear picture where things should be at any time.
- Create code that is easy to change as your application scope evolves.
- Reuse more and more code as you build more applications.

This book roadmap

To achieve our objective we'll follow *a hero's journey* and begin with nothing and work towards a **simple address book application** by the end of the book. We'll begin talking about code conventions and then move to explain the programming pattern called MVC. Each chapter begins with some theory explaining the topic at hand and then moves to a practical section where we apply what we learned to the task of building the address book application.

Even though I am organizing this book as a little journey, you will still be able to cherry pick chapters and learn specific topics out of order since your needs or interests may not be in the same order as presented here. Just be aware that the book sample is built with the book order in mind and code from the middle of the book may depend on code from the early parts.

Code Conventions

The main objective of code conventions is to produce code that is identical no matter who the author is. Code that follow such conventions are immediately recognizable by developers and easier to maintain in the long run. We could say that code conventions are nothing but *codified common sense* which is true in hindsight but may not be clear when all you have is an empty stack and no plan. In this chapter we're going to learn the most common code conventions in our community, and we'll begin with how to name stuff.

How to name things

Names should always be descriptive and meaningful. LiveCode is a very verbose language and you should not be afraid of making your variables and handlers even more verbose if this saves your sanity six months from now. Other programming languages from old times had space and memory constraints that forced the developer to use the smallest names.

I had a computer that used **BASIC** and each variable name could have only up to two letters, so you went from variable **a** to variable **zz** (and cried a lot while maintaining any code).

Many developers that learned to code with similar constraints still name things with cryptic names such as `curcont` claiming that saving some characters in the variable name is good because they type variable names a lot and typing less will lead them to work faster. Naming the same variable with a name like `tCurrentContact` makes your code obvious even to people who have never seen it before and it doesn't require much more typing while presenting many benefits in the long run. We should always name things clearly and avoid using contractions or abbreviations. Don't be afraid to type more characters, the more descriptive something is, the less code commenting you will need to do later. Let's agree on some guidelines for naming things:

- Use meaningful names.
- Use full words and separate them with a capital letter such as `currentContact`.
- Avoid generic names such as `count` and choose more specific ones such as `numberOfApplesInBasket`.

The rules above are great but let's dive deeper and focus on how to name variables.

Naming variables

In the LiveCode community it is very common to use Hungarian Notation⁴ to name variables as explained in the [Fourth World Scripting Style Guide](http://www.fourthworld.com/embassy/articles/scriptstyle.html)⁵. This method of naming variables advocates for the usage of prefixes and suffixes in the variable name to make the variable type and scope clear.

In LiveCode we have global, script local and temporary variables. I almost forgot, we also have constants. Let us mark each scope with a prefix in the variable name:

- **g for global variable:** such as `gApplicationVersion`.
- **s for script local variable:** as in `sDatabaseConnectionID`.
- **t for temporary variable:** like: `tCurrentContact`.
- **p for parameters:** if the variable is passed as a parameter to a handler. Example: `pContactA`.
- **k for constant:** sometimes we also use ALL CAPS and underscores as separators for constants: `kUPDATE_URL`.

Another important caveat is that there is no way to tell if a variable is an array or not in LiveCode just by looking at the variable name, so **we usually use a capital A suffix for array variable names** as in this example: `gAllContactsA` which is a global array variable holding all contacts for a given application.



Tip: Its ok to name a variable `x` or `y` on loops such as `repeat with x = 1 to 10`.

Example:

```
1 global gAllContactsA
2
3 local sCurrentContactA
4
5 command sayHi pContactA
6     local tGreeting
7     put "Hello," && pContactA["name"] into tGreeting
8     return tGreeting
9 end sayHi
```

⁴Check out http://en.wikipedia.org/wiki/Hungarian_notation.

⁵<http://www.fourthworld.com/embassy/articles/scriptstyle.html>

Naming handlers

Some people take a great care when naming their variables and then let all this care go down the drain when naming their handlers. If you think that most of your code ends up as being variable names and handler names, you notice that using some wisdom when naming them will pay well in the long run. Some guidelines for handler names:

- Always have a verb in the name as in `placeFruitInBasket` instead of `fruitBasket`.
- Be descriptive, use names as long as they need to be.

I have a handler called `checkApplicationVersionAndUpdateIfNeeded` and I think it is a great name because it instantly tells me what that handler does.

Example:

```
1 function getContactDataAsJSON
2     // do something
3 end getContactDataAsJSON
4
5 command emailCurrentSelectedContact
6     // do something
7 end emailCurrentSelectedContact
```

We should not underestimate how useful a good name is. Select your verb well and if possible use it as the first word of the handler. As a mental exercise, consider the code below:

```
1 put contacts() into tContactsA
```

From that code there is not much we can infer about what `contacts()` does. Does it accept a parameter? Is it a function to set something or to retrieve something or both? Unless we have comments in the source code we'll need to read the implementation of that function to understand what it is actually doing. This could all be solved by better naming.

```
1 put getAllContacts() into tContactsA
```

Not only we know what it does since it uses the verb **get** but we know it retrieves **all** contacts and not just some selection. Another important naming convention for handlers is regarding their scope. In LiveCode we can have both public and private handlers. If you're building code that is going to be shared, such as a library, then you'd create a public API that is documented and available for others, and a private API that is used only by you. Handlers should not call private handlers of other stacks. And yes, now that we have `private` and `public` keywords there is no way to leak the implementation handlers anymore but still naming them differently helps with maintenance.

Lets use an underscore character before the name of any private handler. So if you have a public command called `saveContact` and this command makes use of the private command `writeContactToFile` then you should named the private handler `_writeContactToFile` like in:

```
1 on saveContact pContactA
2     combine pContactA by cr and tab
3     _writeContactToFile pContactA
4 end saveContact
```

So just by looking at that source code we know that `_writeContactToFile` is a private command. Sometimes, naming something wisely is not enough to convey why something is happening in the source code. For these cases we have source code documentation.

How to document your code

There should be a book dedicated only to teaching how to document code. I suspect developers hate to write documentation because when they think that when they are writing documentation they are not coding. There are some great tools to help you document your application such as [screensteps](http://www.bluemangolearning.com/screensteps/)⁶ but this section is not about documenting your application for your end user (one of the tasks that screensteps excels on) but to create comments alongside the source code to better explain what is going on.

The rule of thumb for code comments is **not to explain what is happening but why it is happening**. The developer can understand what is going on from reading the source code but they may need some help understanding why it should work that way. For example, consider **the poor case of commenting** below:

```
1 function addTwo pNumber
2     // below we add two to a number
3     add 2 to pNumber
4     return pNumber
5 end addTwo
```

That comment is **useless** because it doesn't help the developer at all. Now consider the following better example of commenting:

```
1 function movePlayerDown @pPlayerLongID
2     // We add 48 pixels to the value of
3     // top because our game tiles are 48 pixels
4     // tall, so adding this value moves
5     // the player one tile down.
6     add 48 to the top of pPlayerLongID
7 end movePlayerDown
```

⁶<http://www.bluemangolearning.com/screensteps/>

If at anytime in your source code something is not clear why it should be that way, then it is a good time to write a comment explaining the reasons behind how that code needs to be the way it is. There is no such thing as excessive commenting if all your comments are good quality (mark this affirmation as my personal opinion as this is a controversial topic). The more code conventions and comments you use, the less you will need to remember when you need to fix something in your app in the future.

Tools that use comments for the greater good

There are some great tools out there that use source code comments to build amazing things. There is [lcTaskList](https://livecode.com/extensions/lctasklist/1-3-0/)⁷ that allows you to leave comments with markings such as TODO, FIXME and build reports on the notes you left for yourself. There is also [NativeDoc](http://www.nativesoft.net/products/nativedoc/)⁸ which is a documentation generator for LiveCode that picks comments structured in a specific way and builds HTML documentation from it.

I've also built RevDoc that is like NativeDoc a long time ago, you can check more in [RevUp 64](http://newsletters.livecode.com/january/issue64/newsletter1.php)⁹.

Summary

In this chapter we've learned:

- The importance of good names and conventions.
- That by using prefixes and suffixes in variable names we can convey scope and nature.
- That by being verbose and always using a verb in handlers names we can make our intention (and code) clearer.
- That source code commenting helps developers understand why something is happening and not just how.

The next chapter we'll talk about MVC, a mindset and strategy that is very successful in shipping maintainable code.

⁷<https://livecode.com/extensions/lctasklist/1-3-0/>

⁸<http://www.nativesoft.net/products/nativedoc/>

⁹<http://newsletters.livecode.com/january/issue64/newsletter1.php>

An introduction to MVC

The objective of the [MVC paradigm](#)¹⁰ is to enforce [loose coupling](#)¹¹ between the representation of information and how the user interacts with it, so that you can more easily change one without breaking the other.

By *loose coupling* I mean that your application is divided into units (that in the MVC paradigm we call models, views and controllers) that are not dependent on each other in a hardcoded way. This means separating the way things are stored and retrieved (aka model) from your business logic (aka controller) from the way you display your application (aka view). If you follow this separation, you will be able to change each unit with minimal change to the surrounding units.

To be able to achieve this separation, we should follow [the law of Demeter](#)¹² which states that each of our application units should only have limited knowledge about the other units and restrict its interactions to its closest friends. For example in a complex application with an address book module and a project management module, each module should restrict its interactions to themselves. The project management module should not go changing the data from the address book module without talking to it. The law of Demeter helps us avoid the difficult debugging problem of figuring out where something was changed



Ever caught yourself thinking: “*which part of the code changed that global variable?!*” thats a huge problem in LiveCode (and other languages). People didn’t coin the term [Globals are Evil](#)¹³ for no reason. Be aware of spooky actions at a distance, it scared Einstein and it should scare you too. Try replacing globals with variables with limited scope.

We can think of MVC as a way to untangle our stacks moving from the usual spaghetti-approaching state towards a neat and organized future. Since we’re speaking about stacks, lets draw out some plans to explain how MVC applies to a project. As we progress in this book we’ll build a simple address book example that will be divided into units called *model*, *view*, *controller* and *libraries*. Each of these units is a different stack and will deal with a different problem.

MVC & OOP: The MVC paradigm was created alongside the [Object-Oriented Programming](#)¹⁴ which is one of the main paradigms to approach programing. LiveCode is not an OOP language in the classical sense as understood by SmallTalk developers but that does not mean we can’t leverage what we can from the MVC way of doing things when creating our own applications.

¹⁰<http://en.wikipedia.org/wiki/Model%E2%80%93view%E2%80%93controller>


¹¹http://en.wikipedia.org/wiki/Loose_coupling

¹²http://en.wikipedia.org/wiki/Law_Of_Demeter

¹³<http://wiki.c2.com/?GlobalVariablesAreBad>

¹⁴https://en.wikipedia.org/wiki/Object-oriented_programming

MVC in 20 seconds



The MVC process

The model

Our address book model will be responsible for loading, saving and manipulating address book contacts data. We're going to define an API that will be used by the other units to talk to it and how we implement this contact manipulation is of no concern to them. We could even have more than one model for the same data type but with different implementation, such as a model that saves to a database, a model that saves to text files, a model that saves to a remote server. As long as we implement the same public API, we can change the internals of the model without touching the other units. In the classic MVC implementation, the Model will notify interested parties if the data changes.

The controller

The controller is a stack that holds your business logic. While the model knows how to manipulate contact data, it doesn't know why or when it should manipulate it. The controller is responsible for

talking to the model, like a transit cop directing traffic to and from the model to the other units.

In the classic MVC implementation, the controller is responsible for getting input from keyboard and mouse or whatever input gizmo you are using and decide what to do with it. It talks directly to the model telling it what it should do the data. It can also talk directly to the view since some of the input may require stuff in the view to change.

The view

The view is the presentation layer with your cards and controls. The user sees and interacts with the view and the view talks only to the controller which in turn talks to the model.

This way you can try different views and layouts without touching your business logic because that is located on the controller. When you're building cross-platform applications, you can build multiple views to create interfaces that are suitable for the various form factors you're deploying to, like desktop, tablets and smartphones, all from a single application source.

In traditional MVC implementations, the model use broadcasting techniques to inform the views when data changes. In this loose coupling, the model has no idea what the view needs to change, it is just broadcast to all interested parties that they should act upon data change.

Libraries

It is good practice to build code that is reusable beyond your current application development. Libraries due to not being tied to the business logic of your application, are the most reusable pieces of our softwares, the more generic looking code that you can build into libraries the easier your development will be in the future.

As an example, suppose you have a function that picks a name like *andre alves garzia* and converts it to TitleCase like *Andre Alves Garzia*. This function is useful for the address book application and many other applications, so it could be added to a library and reused in other projects.

Summary

This was just an at a glance introduction to MVC. Each chapter now will dive deeper into one of these concepts starting with models. By each chapters end, we'll have one unit ready and be closer to the finished address book application. This small chapter had a bunch of links in it, you really should check them all.

Now, lets see what models are made of.

Models

Models are the collection of routines that deal with the manipulation of some domain specific data used in your application. Not all applications need models, you can build a calculator without adding any model. Models are most useful when you're dealing with complex data or when you need to keep track of stuff between sessions of your applications. If you need to save and restore data in your application then you will benefit from using models.

What goes into the model and what doesn't

Models should contain all code needed to manipulate some domain specific data used by your business (I am repeating myself intentionally, to fix this message). In the case of an address book application, our model will handle contact information data and have all the routines needed to create, delete, edit, update and search for contacts. Our model will deal with all that but it won't know how to display a contact on the screen or what combination of mouse and keyboard is used to edit them. Models are accessed by their public API which is the collection of handlers that a controller use to manipulate them.

Building a sane API

It is easy to conclude that **models are as good as their API**. You can have the most elegant file representation for an address book contact but if the routines that the controller will use are poorly designed, then you'll end up adding more friction to your code maintenance chores.

A **good API design does not let implementation details leak**. For example, suppose we want to save a contact in the address book, the controller could use code similar to:

```
1 on mouseUp
2     put getContactFromCardControls() into tContactA
3     openAddressBookContactFile
4     addContact tContactA
5     saveAddressBookContactFile
6 end mouseUp
```

Even though we can clearly understand what is going on, **this is not a good design because it leaks all the implementation details**. The controller doesn't need to know about files nor does it need to remember to open and save them. This is responsibility of the model. A better API would be:

```
1 on mouseUp
2     put getContactFromCardControls() into tContactA
3     saveContact tContactA
4     if the result is not empty then
5         // handle error
6     end if
7 end mouseUp
```

The whole business with files is left to the model implementation. The controller should not care if the contact is saved to disk, to a database, to a remote server, to a shared drive or printed. All it needs to do is request a contact to be saved and check the result of such call.

With that in mind, it is time we get practical and build our model.

Practical: Building an address book model

This book is called *LiveCode Advanced Application Architecture* and not *The Art and Science of Contact Management* so our model will be a bit naive. I am more interested in showing you a workflow and strategy for LiveCode software development than creating a bullet proof address book implementation. We could turn this book into a 900+ pages behemoth if we started going down that rabbit hole, so we'll keep things simple.

Many developers will start to think about models by thinking on how data will be stored and then creating the API on top of that. I don't like this approach, I prefer creating the API the controller will call first and then worry about how to implement it. This way we can have an elegant API first and change implementation as needed.

The public API

First lets figure out what our needs are. We need to be able to manipulate contacts and keep track of them. So our model needs to be able to:

- Create a new contact.
- Return all contacts.
- Return a specific contact.
- Search for a contact.
- Save a contact.
- Delete a contact.
- Update a contact field.

Lets look at our public API:

Type	Name	Params	Purpose
function	createNewContact	pFieldsA	Creates a new contact based on the fields passed
function	getAllContacts		Returns an array of all available contacts
function	getContactByID	pID	Return a specific contact
function	filterContactsByField	pField, pValue	Return an array of contacts by searching a field
command	saveContact	pID,pContactA	Saves a contact
command	updateContactField	pID, pField, pValue	Updates a field in a contact
command	deleteContact	pID	Deletes a contact

This public API is all the controller will use to implement our Address Book application. The inner working of the model will be of no concern to both the controller or the view. Since our model will not depend on anything besides itself, we will be able to reuse it in other future projects. The collection of handlers above is the only public API for our model, all the other handlers are private. Lets explain the rationale behind each handler before we move onto the implementation.

Contact Model: Creating the stack

Create a new script-only mainstack in LiveCode and call it `model.contact` with its filename called `model.contact.livcodescript`. This is not a convention followed by others as far as I know. In this book we'll use the following files:

Role	Stack name	File name
Model	model.contact	model.contact.livcodescript
Controller	controller.addressbook	controller.addressbook.livecode
View	view.addressbook	view.addressbook.livecode

All our implementation will go into *the stack script*. Our address book application will use the model as a [Library Stack](#)¹⁵ so that its public API is available on the message path. By keeping the model on its own file, we'll be able to reuse it easily.

Contact Model: the data storage

To keep the implementation simple, our model will work with files. We'll use a single file called **mycontacts.data** to hold our data. We could make a more robust address book by using a SQL database or even an online storage but we're going to keep things simple. Our implementation we'll be as simple as possible.

This `mycontacts.data` file will just be the dump of an array into disk. We'll write and read from it as needed.

¹⁵<http://livecode.com/developers/api/6.0.2/command/start%20using/>

Warning: If you plan to ship software using the iTunes App Store or the App Store for macOS then you need to know that Apple is very strict on what you can write to disk and where. Personally I think that the App Store requirements are hurting developers but that is a different matter altogether. Just remember to learn the requirements before trying to get your software approved by Apple.

Contact Model: the public API

We'll implement the public API as if the private handlers we need existed already then we'll implement them afterwards. I call this **the naive dreamer approach** where you dream you have the handlers you need and then later you're forced back into reality and need to implement them all yourself (also known as: the principle of eventual implementation).

Creating a new contact

This handler is a function because it needs to return a value which is the ID of the created contact or false. We could use the `result` for this but it makes it harder to compose with other statements such as `if` clauses. This function will receive an array where each key and value pair is a field for the contact entry. There will be no checking against a schema or field validation, whatever goes inside the array will become the user record.

This function will be called by the controller when it wants to add a new contact. There are potential sources of difficulties on this function, more specifically, it needs to make sure that the ID generated for a given user will not collide with any other user. This is easy in a single threaded application such as a normal LiveCode based application but if we start thinking in terms of networked computers accessing the same contacts database, then, new safeguards must be put into place.

To ensure the atomicity of the `createNewContact` operation, the function will make sure that the data is saved before it returns control to the calling handler.

```
1 function createNewContact pFieldsA
2     put _getAllContactsFromFile() into tAllContactsA
3     // below is a poor man unique id to be used as primary
4     // key for the new contact. This is not the correct way
5     // of doing this, but it is good enough for a book sample.
6     put the millisecs & random(999) into tUniqueID
7     // the code below checks to see if the key generated above
8     // is already in use and will keep generating keys until it
9     // generates something that is not currently in use.
10    repeat while the keys of tAllContactsA[tUniqueID] is not empty
11        put the millisecs & random(999) into tUniqueID
```

```
12     end repeat
13     put pFieldsA into tAllContactsA[tUniqueID]
14     _writeContactsToFile tAllContactsA
15     return tUniqueID
16 end createNewContact
```

This code loads the data from the file, generates a unique id, adds the new contact to the array of contacts and finally saves the file. Your handlers should always be as short as possible. Rule of thumb is that if your handler is taller than your screen then you're doing something wrong and should refactor into multiple entries.

Warning: That unique id on that code is all but unique. That's not the correct way of generating ids. It's a naive implementation just to get us going. The wonderful Mark Smith made a [Unique ID library available on his site](http://marksmith.on-rev.com/revstuff/)^a, if you need UUIDs, you need this library.

^a<http://marksmith.on-rev.com/revstuff/>

Our function returns the unique id for that contact so that we can refer to it in other handlers.

This implementation leaves two private handlers for us to implement later: `_getAllContactsFromFile()` and `_writeContactsToFile`.

Getting contacts

Let's implement both `getAllContacts()` and `getContactByID` because their code is very similar and straight forward.

```
1 function getAllContacts
2     put _getAllContactsFromFile() into tAllContactsA
3     return tAllContactsA
4 end getAllContacts
5
6 function getContactByID pID
7     put _getAllContactsFromFile() into tAllContactsA
8     return tAllContactsA[pID]
9 end getContactByID
```

Be aware that our routines are as atomic as possible. They tend not to rely on stuff such as files being open or a variable in memory having all the contacts. Every time we need to check the contacts we're loading them again from disk. This we guarantee that the information on the file is always what you're dealing with, this lowers our chances of having a state where the data in memory is out of

sync with the data on the disk, but that causes more file activity and latency then needed. If we were dealing with thousands of records, we would need to implement things differently.

There are situations where you don't want to hit the disk that often. You need to know which solution is better for your case.

there is no new private handler here so lets move on.

Filtering contacts by field

Filtering and searching are among the family of tasks that are usually better handled by a SQL-based database. Our address book will be really simple and will allow just filtering by a single field at once.

There are whole books, master degrees, PhD thesis, on the topic of filtering and searching. If you have complex searching needs¹⁶ then I suggest you learn more about patterns such as [Map and Reduce](#)¹⁶.

```
1 function filterContactsByField pContactListA, pField, pValue
2   put _getAllContactsFromFile() into tAllContactsA
3   repeat for each key k in tAllContactsA
4     if pValue is in tAllContactsA[k][pField] then
5       put tAllContactsA[k] into tContactMatchesA[k]
6     end if
7   end repeat
8   return tContactMatchesA
9 end filterContactsByField
```

Again, another straight forward implementation. We load all contacts, loop them checking if pValue is inside the given field and return the matches.

Saving a contact

```
1 command saveContact pID, pContactA
2   put _getAllContactsFromFile() into tAllContactsA
3   put pContactA into tAllContactsA[pID]
4   _writeContactsToFile tAllContactsA
5 end saveContact
```

Its very similar to our createNewContact() function but since we already know the unique id for that contact we don't need to compute it.

All functions that returns a collection of contacts have the unique IDs as the first level of the returned array. So you will only call the saveContact command if you already know which contact you want to save (by knowing its unique id).

¹⁶<http://en.wikipedia.org/wiki/MapReduce>

Updating a contact field

This is just a convenience handler. The `saveContact` command could be used for all update needs. I am showing you this implementation just to demonstrate that your public API handlers can build upon other handlers from the public API itself.

```
1 command updateContactField pID, pField, pValue
2   put getContactByID(pID) into tContactA
3   put pValue into tContactA[pField]
4   saveContact pID, tContactA
5 end updateContactField
```

Contact Model: the private API

To finish our model we need to implement two private calls: `_getAllContactsFromFile()` and `_writeContactsToFile`. These handlers are not supposed to be called by any routine outside the model stack. We could replace the whole implementation of that stack with a SQL-based solution maintaining the same public API calls and the controller and view wouldn't even notice.

Loading contacts from file

```
1 private function _getAllContactsFromFile
2   // We need to preserve the current default folder
3   // The rest of the app may depend on it and we don't want
4   // side effects bugs playing us.
5   put the defaultfolder into tCurrentDefaultFolder
6   set the itemdel to "/"
7   set the defaultfolder to item 1 to -2 of the effective filename of this stack
8   if there is a file "mycontacts.data" then
9     put url "binfile:mycontacts.data" into tFileContents
10    put arraydecode(tFileContents) into tAllContentsA
11  else
12    put empty into tAllContentsA
13  end if
14  set the defaultfolder to tCurrentDefaultFolder
15  return tAllContentsA
16 end _getAllContactsFromFile
```

There are two possibilities for our routine. Either there is a file holding the contacts in which case we need to load and return them or there isn't a file with contents and we can just return empty.

One important thing in this implementation is our attention to avoid [side effects](#)¹⁷. Side effects happen when a handler does something that is outside its domain specific needs.

For example, suppose you are building an application that uses our contact model and is reading and writing to the Desktop folder by setting it as the default folder. If our `_getAllContactsFromFile()` doesn't change the default folder back to what it was before it was called then when that function returns, the default folder would have changed to the same location as the model stack. A developer doesn't expect a function called `_getAllContactsFromFile()` to have the *unintentional and undocumented* effect of changing the default folder.

That's why we keep track of what folder was set as default and remember to set it back before returning. If we keep our handlers nice and tidy, we minimize the risk of unintentional side effects.

Writing contacts to file

```
1 private command _writeContactsToFile pAllContactsA
2   // We need to preserve the current default folder
3   // The rest of the app may depend on it and we don't want
4   // side effects bugs plaguing us.
5
6   put the defaultfolder into tCurrentDefaultFolder
7   set the itemdel to "/"
8   set the defaultfolder to item 1 to -2 of the effective filename of this stack
9   put arrayencode(pAllContactsA) into url "binfile:mycontacts.data"
10  set the defaultfolder to tCurrentDefaultFolder
11 end _writeContactsToFile
```

The code to prevent side effects is longer than the actual code to write the contacts to a file.

With that final command we end our model implementation. That's all we need for a basic address book app. We may refine the code as new needs arise along the book.

Summary

In this chapter we had a lot of work. We learned about models and what they are used for. We also implemented a simple contact model to be used by our address book sample stack.

In the next chapter we'll build our controller.

¹⁷http://en.wikipedia.org/wiki/Side_effect_%28computer_science%29

Controllers

Controllers are like transit cops

The main responsibility of the controller is to organize the business logic and the flow of our software. Like traffic cops, the controller will stay in the very middle of your app orchestrating the flow of data and events between the different units. This is the unit where you should pour most intelligence and attention because a good controller design will help experiment with variant models and views with ease.

Just like traffic cops have their own attributions and you don't expect them to act like detectives or the swat, controllers have their own tight scope. It is time for us to learn what controllers are made of.

What goes into the controller and what doesn't

The controller is where you solve your problem. Each software is trying to solve a problem or accomplish something. In this task, it uses different models, networked resources and user interfaces, many of which are also used by other softwares in the same problem space. The *unique selling point* of your software, that spark that makes your solution special, that is what goes into the controller.

Disclaimer: Yes, you can have software which the unique selling point is its fancy alien UI or software that has some clever and elegant file structures which make it more useful than the alternatives. This is all good and if you come to realize that your special thing is not how you solve a problem but what lies elsewhere in your structure, then so be it. Still, most software will have its special thing in *how it solves a problem*.

To use a culinary analogy, let's think about pizza. Most pizzas use the same ingredients, which are analogous to our models. Pizza ingredients don't know they belong into the pizza, they could be used for other recipes. Some pizzas are beautifully arranged and looking at them is almost as good as eating them. Those pizzas are like software with very pretty UIs. Still, the pizzas you tend to remember the most and come back to, are the ones that tend to be prepared in a special way. This preparation step, where a master pizzaiolo takes the common ingredients and makes them into something special, that is what a good controller is. Everyone loves problems that are solved in an elegant and tasty way. Specially if your problem is craving for a pizza at 1h27 AM like I am right now.

A good controller deals with some [domain specific problem](#)¹⁸ you're trying to solve. If your app deals with more than one domain specific problem such as an app that deals not only with contacts but also with project management, then, you will end up with more than one controller (and also more than one model).

Controllers and views are the least reusable part on an app because they are the business logic and the specific UI to apply that logic. Models and libraries are the most flexible parts which tend to be used over and over. If you end up reusing not only them but also controllers and views in a different application, then you're basically embedding an app into another, that is not a bad thing but its not the kind of reuse that we're talking about here.

What shouldn't be on a controller

Controllers will exchange data with the views and the models by calling their public APIs, because of this they should not make any assumption about anything in your app besides such APIs. They can't reference controls or files or anything that should belong into one of the other units.

If you're wondering if some code belongs in the controller or model, then, ask yourself these questions:

- Is this code something that I might use in another application?
- Does this code relates to my business logic or it is purely some model manipulation?

If the answer to any of these questions is *YES* then you should probably implement that in the model layer. Similarly, if any of the code you're writing on the controller (or the model, or the view) is so generic that you could reuse it elsewhere, then place it on a library (more about that later in this book).

A controller never assumes anything about the views, it just call their public implementation. The views themselves also call the public methods from the controller, the less coupling or dependency we have on hard-coded stuff the better.

Let's get practical!

Practical: Building an address book controller

In the MVC chapter, we've talked about *Loose Coupling*, if you haven't read that or don't remember it, please go back to that chapter and check it out because that is crucial for understanding how and why we're going to implement our controller in a specific way.

There are many ways to achieve *loose coupling* in LiveCode, basically you can overengineer or underengineer as much as you can and still end up with something usable. In this section, I am

¹⁸https://en.wikipedia.org/wiki/Problem_domain

going to introduce a common OOP pattern called [publish and subscribe pattern](#)¹⁹, this is going to be used by both the controller and the view and it is a very elegant way to achieve independence between our units.

The publish and subscribe pattern

This is a messaging pattern which the main objective is to exchange data between *publishers* and *subscribers*. The beauty of it is that the *publishers* don't know who the *subscribers* are, they just broadcast messages, and those are picked automatically by the subscribers.

[Wikipedia page for the publish/subscribe pattern](#)²⁰: “In software architecture, publishâ€subscribe is a messaging pattern where senders of messages, called publishers, do not program the messages to be sent directly to specific receivers, called subscribers, but instead categorize published messages into classes without knowledge of which subscribers, if any, there may be. Similarly, subscribers express interest in one or more classes and only receive messages that are of interest, without knowledge of which publishers, if any, there are.”

By using this pattern our controller will at the same time subscribe to some messages and broadcast others, our view will do the same. They will not be aware of each other but messages will flow from one to the other.

Using the publish/subscribe pattern in LiveCode

There are multiple implementations of this pattern in LiveCode. I am bundling a simple one I made with the source code of this book, this stack is going to be used as a library stack, it is called `aagPubSubLib.livecode`. So, I will not explain how we implement the pattern itself here, you can check the source code if you want, it is open. I'll explain how to use it though.

That library public API has four handlers:

subscribe command

This command is used to register your interest in a specific message. It accepts four parameters: the message, the callback for when it is received, the target for the callback and an option flag.

¹⁹https://en.wikipedia.org/wiki/Publish%E2%80%93subscribe_pattern

²⁰https://en.wikipedia.org/wiki/Publish%E2%80%93subscribe_pattern

```
1 on mouseup
2     subscribe "search contacts", "_searchContacts"
3 end mouseup
```

In the example above, we're registering interest in the message *"search contacts"* and telling that when such message is broadcast, that the handler called *"_searchContacts"* should be called. By default, if the third parameter is empty, then the caller of `subscribe` is marked as the target for the callback. As for the first parameter, we can name it whatever we want as long as both the publisher and subscriber use the same name (or they won't find each other when the broadcast happen).

There is an analogous `unsubscribe` call that removes the callback.

broadcast command

Use that to broadcast some message. All scripts that registered interest in that message will be activated.

```
1 on mouseup
2     broadcast "search contacts", "andre garzia"
3 end mouseup
```

The example above would trigger the `_searchContacts` call since it subscribed to the *"search contacts"* message.

Implementing our controller

That is all we need to start implementing our controller, we'll define some messages that the controller broadcasts and some messages that it is interested in (which will be broadcast by the view). We'll also add some housekeeping routines to start the view and make our life easier. In this book example, we'll have only one controller and if we were building a standalone, that would be our mainstack, so some initialization routines that would usually go into that stack are being placed into the controller. Since our controller is our main stack, we're going to make it a normal LiveCode stack or it won't be able to hold *standalone settings data* from the IDE. You can't build standalones from script-only stacks.

Our controller will be called `controller.addressbook.livecode`.

Messages it is interested in

Again, let's pretend that the messages exist and devise what we want our view to send us.

Message	Callback
search contacts	_searchContacts
update contact	_updateContact
delete contact	_deleteContact
save new contact	_saveNewContact
list contacts	_listContacts
get contact	_displayContact

On the left column, we have the messages we're going to subscribe. Those are the events that the *view* will send in the *controller* direction.

Messages we're going to broadcast

Message	Params
display contacts	tAllContactsA
display contact	tContactA

These are the two messages that our controller broadcasts. Orders to display the list of contacts or a specific one.

Intermission

Now, lets do a health check and check if you, dear reader, are still with me. Publisher and Subscriber patterns are not something we usually see in most stacks in the LiveCode community and the main reason why I added *advanced* to the title of this book.

If you're having trouble wrapping your head around it. Don't worry, those things are complex. Checking the source code for both the *controller* and the *view* will help a lot. To summarize using other words, publishers are broadcasting events and subscribers are listening to them. When an event happens (aka is broadcast) the callbacks in the listeners will trigger. The advantage over the normal LC message path is that a single event might trigger multiple callbacks across different stacks. So different parts of your application might be listening for the "*list contacts*" message, if it happened, then, they would all receive the event.

For our simple demo stack, this might sound overkill but once you start thinking in those terms, it becomes a very useful pattern that leads to code that is easy to maintain.

The controller script

Our controller has no UI even though it is a normal LiveCode stack. In the version bundled on this book, I've added label fields and a description but all that is useless. All the code for the controller is in the *stack script* and thats what we're going to discuss here.

Once the *controller stack* opens, it needs to setup the *subscriptions* for the messages it want and get out of the way by showing the view. So our `preOpenStack` becomes:

```
1 on preOpenStack
2   _initializeController
3 end preOpenStack
4
5 private command _initializeController
6   start using stack "aagPubSubLib"
7   start using stack "model.contacts"
8
9   _setupEventSubscriptions
10  _launchView
11
12  if the environment is not "development" then
13    hide this stack
14  end if
15 end _initializeController
```

I've decoupled the `preOpenStack` from the *controller initialization* to make our life easier in case we decide in the future that the *controller* needs to be initialized using some different method. The first lines of the code just put our model and publisher/subscriber pattern library into use, after that it does exactly what we mentioned above, it subscribes to the messages it is interested in, display the view and then get out of the way. Of course, if you run this code it will fail because we don't yet have the two private commands used, lets build them.

```
1 private command _setupEventSubscriptions
2   subscribe "search contacts", "_searchContacts"
3   subscribe "update contact", "_updateContact"
4   subscribe "delete contact", "_deleteContact"
5   subscribe "save new contact", "_saveNewContact"
6   subscribe "list contacts", "_listContacts"
7   subscribe "get contact", "_displayContact"
8 end _setupEventSubscriptions
9
10
11 private command _launchView
12   go stack "view.addressbook"
13 end _launchView
```

You might notice a very tight coupling there, our *controller* is assuming the name of our *view stack*. We could solve that with more messages (like having a *show view* message) but to be

honest, some coupling is OK. The controller is already putting that stack into use anyway in the `_initializeController` command.

Now, all that is left is implementing the callbacks for all those subscribe calls. As you can see, I've created the messages mimicking the model API (mostly because I am lazy), so each callback will be straight forward. Our commands will communicate back to the view using broadcast calls.

```
1 command _listContacts
2   put getAllContacts() into tAllContactsA
3   broadcast "display contacts", tAllContactsA
4 end _listContacts
```

The `_listContacts` command shows a pattern we're going to reuse in the other handlers. We call some routine from the *model*, wrap the results into a variable and *broadcast* it.

```
1 command _displayContact pContactID
2   put getContactByID(pContactID) into tContactA
3   broadcast "display contact", tContactA
4 end _displayContact
```

In the commands below, we're showing composition by making them rely on the command above.

```
1 command _saveNewContact tContactA
2   put createNewContact(tContactA) into tContactID
3   _displayContact tContactID
4 end _saveNewContact
```

```
1 command _updateContact tContactA
2   saveContact tContactA
3   _displayContact tContactA["id"]
4 end _updateContact
```

There are some extra handlers there to deal with contact deletion but it is quite similar to what we already have here. Check out the *stack script* of the bundled source for the complete script.

Summary

In this chapter we dived deep into *controllers*, arguably the most important part of an application, the one thing tying it all together. We also learned about a very useful pattern called *publisher/subscriber pattern* and how to leverage it from LiveCode. We're now ready to build our *view*.

Views

I am tempted to say that “*views are the most visible part of your application*” but I guess that would be too cheesy. Yet, it is true. Views are what the end users will interact with and what they will grow to assume is the whole of your application. Software is like an iceberg, your models, libraries, controllers, all belong to the large piece of ice under the sea while your view is the small piece visible above. Users see that piece above the sea and assume that's the whole thing without giving any notice to the vast realm below.

Crafting good user experiences is not only a real science but it is also, in my opinion, an art. It is very hard to convey how to do it in a small chapter like this but I will give you some advice on common tips and tricks. Still, if you want to dive deeper into usability and user interfaces there is a ton of information and courses online.

Practical tips for interface design

- Don't make your software a scavenger hunt. If the options available for your user are hard to find or require some complex interaction for them to work, you're adding friction to the usage of your software which will lead to more support requests and potentially more bugs as the software is used in the wrong way.
- Group features related to the same domain together. If you have a modal dialog with a button to accept some change or to cancel them, those buttons should be near one another so that when the user focuses their attention to that area, they perceive all options available to them.
- Reduce the amount of clicks needed to navigate inside your software. Try to achieve the “under 3 clicks” dream which makes the user go from any place in your application to any other place in it under three clicks. It is an arbitrary number but if it takes your user 10 clicks to navigate from one part of the application to another there is a chance they will never go there.

What goes into the view and what doesn't

In some languages, *views* don't have any code in them, they are this *dead thing* and the controller does everything. I think this is a waste of potential and I think *views* should contain the necessary code for them to be able to execute their work of being the user interface to your application. For example, consider a financial application that is listing deposits and withdraws from a bank account. Suppose you want to color withdraws in red and deposits in blue. The code to do that can be in the view as it doesn't affect the controller or the model, it is just cosmetics added to provide a better user experience. It is not useless but doesn't affect or rely on stuff from the other units, so, it is OK to add that code to the *view*.

In LiveCode it is very tempting to start adding code to views, as in, lets build a whole feature inside a `mouseUp` handler, it is so easy! Then, when it is time to build a different view, or trigger that feature from some other control, we end up with calls such as:

```
1 send "mouseUp" to button "myButton" of card "that distant card" of stack "some compl\
2 etely different location"
```

If you're building a feature, place it a model, library or controller depending on the nature of it unless it is something that relates only to the view.

Multiple views are the key to cross-platform applications

The main benefit of our *loose coupling* and the *"publisher/subscriber pattern"* we used in the previous chapter is that there is nothing preventing us from having multiple views. We could have `view.addressbook.mobile` and `view.addressbook.desktop` with the *controller* deciding which one to launch depending on the device being used and everything would still work. You could even launch both of them at the same time and they'd work in tandem because they would both subscribe and react to the same messages.

Building multiple views makes it a lot easier to ship cross-platform code from a single codebase and is much easier than using a single view and `resizeStack` handlers to fix everything at runtime.

We're going to build a single *view* considering only Desktop usage but there is nothing preventing you from building a second one for mobile. It is an awesome exercise actually and you should totally do it.

Practical: Building views for the address book application

For the *view* we'll take a different approach. If I guide you into building the whole view, we'll waste a ton of time. Lets just open the bundled stack and I will describe how it works, it is much better. The *view* file is called `view.addressbook.livecode` and it looks like:



The contacts list view


The best way to follow this chapter is to open the bundled controller called `controller.addressbook.livecode`. By opening it, you'll end up with the library, model, controller and view into memory.

How our view work?

Our *view* is a stack with two cards. One card will be used for listing contacts, this card is called *list view* and another card called *details view* that is used for viewing and editing a single contact details. Unlike our controller, in which we placed all the logic into the *stack script*, for the *view* we're going to make use of both *card scripts*.

When a card opens, it will *subscribe* to the messages it needs and when it closes, it will unsubscribe to those messages thus keeping our subscription lists nice and tidy. Each card will only know about the task it needs to do, so, from the *list view* card, when the user wants to *edit a contact*, a call will be dispatched for the *details view* card. Each control in the cards have just the minimal amount of code to call a handler on the *card script* to do whatever needs to be done. There is no business logic in the view, just calls from the controls into the *card scripts*, which will do the message exchange with the *controller* using the *publisher/subscriber* routines.

The contacts list



The screenshot shows a window titled "Address Book" with a standard macOS-style title bar (minimize, maximize, close buttons). Below the title bar, there is a search section with a label "filter:" followed by a text input field and a "Search" button. The main area of the window contains a table with two columns: "full name" and "email". The table has a vertical scrollbar on the right side. The first four rows of the table contain the following data:

full name	email
Andre Alves Garzia	andre@andregarzia.com
John Doe	johndoe@example.com
Bruce Wayne	batman@gothan.com
Mr Fox	fox@example.com

At the bottom right of the window, there are two buttons: "Add new" and "Edit".

The contacts list view

This card has the following controls:

- a *label field* for the search feature.
- an *input field* for holding the search terms.

- a *search button* for triggering the search feature.
- a *datagrid* for listing contacts.
- an “*add new contact*” *button* for adding new contacts.
- an “*edit contact*” *button* for editing the selected contact in the DataGrid.

If you check the scripts in any button or field, you will find single line scripts calling into the *card script*, which is the interesting part. Lets go step by step into it.

```
1 on preOpenCard
2   _initializeView
3   broadcast "list contacts"
4 end preOpenCard
```

On `preOpenCard` we `_initializeView` and broadcast a message asking for the contact list (which is handled by the *stack script* of the *controller* under the `_listContacts` handler).

```
1 command _initializeView
2   subscribe "display contacts", "_displayContacts"
3
4   dispatch "resetData" to group "contact list"
5 end _initializeView
```

The `_initializeView` is very straightforward, it subscribes for the only message that this card is interested about, which is *list contacts* and it also clears the DataGrid so that it doesn't contain leftover data from some previous interaction.

```
1 on closeCard
2   unsubscribe "display contacts", "_displayContacts"
3 end closeCard
```

`closeCard` is all about tidiness, it unsubscribes to that message.

```
1 on searchContacts
2   put field "search terms" into tSearchTerms
3   if tSearchTerms is not empty then
4     broadcast "search contacts", tSearchTerms
5   else
6     broadcast "list contacts"
7   end if
8 end searchContacts
```


Searching is more complex because it needs to take into account if the user cleared the search field, thus causing all the contacts to be displayed. Searching and listing contacts are two different messages, when they are broadcast, they are picked up by the appropriate handler in the *controller stack script* — `_listContacts` or `_searchContacts` — which in turn broadcast the *display contacts* message passing all contacts or the result from searching. So from the point of view of the DataGrid, both cases are nothing but *display contacts*, it is not aware of the search feature, it is completely decoupled from it.

```
1 on _displayContacts pContactsA
2   go card "list view"
3   set the dgData of group "contact list" to pContactsA
4 end _displayContacts
```

For the final handler, which is the one that actually displays the list of contacts, we first make sure we're in the right card and then set the data onto the DataGrid.

You might be wondering where is the code for the *Add New* and *Edit* buttons, but those buttons send events to the *details view* card, their logic is not handled by the *list view* card. This is to enforce separation of concerns. Only the *details view* card know about how to add or edit contacts, all the *list view* knows is how to list contacts.

The details view



The screenshot shows a window titled "Address Book" with a standard macOS-style title bar (minimize, maximize, close buttons). The window contains a form with the following fields and labels:

- Full Name**: A text input field containing "Bruce Wayne".
- E-mail**: A text input field containing "batman@gothan.com".
- Phone number**: An empty text input field.
- Address**: A large, empty text area.

At the bottom of the window, there are three buttons: "Delete", "Back", and "Save".

The details view

This card presents a collection of fields holding the contact information. There is no logic in the input or label fields. The buttons *Delete*, *Back*, *Save* have the minimal necessary code to call into the *details view card script* which is the script that holds all the logic. Lets go over it.

```
1  local sContactID
2
3  on preOpenCard
4      _clearUI
5      _setupEventSubscriptions
6  end preOpenCard
7
8  command _clearUI
9      repeat for each item tField in "full name,phone,email,address"
10         put empty into field tField
11     end repeat
12 end _clearUI
13
14 on closeCard
15     _removeSubscriptions
16 end closeCard
17
18 private command _setupEventSubscriptions
19     subscribe "deleted contact", "_deletedContact"
20     subscribe "display contact", "_displayContactDetails"
21 end _setupEventSubscriptions
22
23 private command _removeSubscriptions
24     unsubscribe "deleted contact", "_deletedContact"
25     unsubscribe "display contact", "_displayContactDetails"
26 end _removeSubscriptions
```

Let's analyse a longer chunk of the script this time as this is probably getting quite comfortable for you. On `preOpenCard` we clear the user interface to remove any leftover data from previous interaction and also subscribe for the messages we want while still being careful to unsubscribe from them if we move to another card.

The most interesting part is the use of a *script local variable* to hold the ID of the contact we're currently editing. There were other ways of solving this but this was quick and won't cause too much trouble as it can only be changed by the *card script*. This variable is also used to figure out if we're adding a new record or editing an existing one. If `sContactID` is empty then, it is a new record.

```
1  command addNewContact
2      go card "details view"
3      put empty into sContactID
4  end addNewContact
5
6  command editContact pID
7      go card "details view"
8      put pID into sContactID
9      broadcast "get contact", sContactID
10 end editContact
```

As explained above, editing or creating a new record is a matter of holding the ID of an existing record in `sContactID` or leaving it empty. The commands above are the ones called by the buttons on the `list view` card. In the case of `editContact` we also need to broadcast a message requesting the contact details. We could get those details from the `DataGrid` itself but to ask for them again is good practice as we could have built this in a way that the `DataGrid` only receive a subset of each contact data.

```
1  command _displayContactDetails pContactA
2      go card "details view"
3
4      repeat for each item tField in "full name,phone,email,address"
5          put pContactA[tField] into field tField
6      end repeat
7
8      put pContactA["id"] into sContactID
9  end _displayContactDetails
```

The command above is the callback triggered from a broadcast from the *controller stack*. When our *view* asks for a contact information using the *get contact* message, the *controller* responds by broadcasting a *display contact* message which ends up triggering this command.

```
1  command saveCurrentContact
2      repeat for each item tField in "full name,phone,email,address"
3          put field tField into tContactA[tField]
4      end repeat
5
6      if sContactID is a number then
7          put sContactID into tContactA["id"]
8          broadcast "update contact", tContactA
9      else
10         broadcast "save new contact", tContactA
```

```
11     end if
12 end saveCurrentContact
```

Saving a contact is just a matter of assembling an array with the data and broadcasting the appropriate message. There is a bit of code there to double check if we're editing an existing record or adding a new one.

All that is left is deleting records. The code below takes care of it. There is a broadcast to delete a contact and a callback so that we know it worked.

```
1  command deleteCurrentContact
2      if sContactID is not a number then
3          go card "list view"
4          exit to top
5      end if
6
7      broadcast "delete contact", sContactID
8      go card "list view"
9  end deleteCurrentContact
10
11 command _deletedContact
12     answer info "Contact Deleted!"
13     put empty into sContactID
14     go card "list view"
15 end _deletedContact
```

There you have it! Congratulations! You now have a complete MVC-based LiveCode application.

on closeBook

This is the end of the book, I know I want more too. In the next months I plan to do monthly releases of this book, not only because I am sure I will need to issue erratas, but also because there is much more to cover. This was the minimal set of features and discussion I felt was needed to move a beginner LiveCoder into a more seasoned developer experience.

By reaching this point of the book and examining the bundled source code, you should be comfortable with the following things:

- **Naming anything** under the sun! You should be an expert namer now, and both your variables, stacks, and handlers should all use descriptive and meaningful names, all cohesive with the development experience you're building.
- **How to document your code** to save precious time in the future as you forget how stuff was made in the first place and, or, need to on-board more developers into your project. The reasons behind your coding choices should now be apparent not only in the code itself but in the accompanying code comments.
- **Splitting your code into the MVC pattern** making all the units loose coupled to each other by leveraging the message path and useful patterns such as the *publisher/subscriber* pattern.

This has been a quick journey but I hope it was worth it for you, I know I am sure it has been worth for me. Please, don't hesitate to reach out to me with comments, feedback and request for new topics or clarification. My email for this book is feedback@andregarzia.com²¹.

Don't forget to check everything I am building for LiveCode, the entry point for all my creations is: <https://andregarzia.com/livecode>²².

See you all around! Andre

²¹<mailto:feedback@andregarzia.com>

²²<https://andregarzia.com/livecode>